
Introduction

This document contains a short hands-on tutorial that is aimed to illustrate many of the facilities available within Mobile BASIC. If you have never used Mobile BASIC before then please start with the Getting Started Manual.

Variables and Expressions

In common with other languages, Mobile BASIC has several technical terms that you may encounter. 

The term LVALUE is used to refer to a value that has a named location in memory, for example, a variable name or array element. An lvalue can be assigned a value. A constant or an expression is not an lvalue since you cannot assign a value to them. An LVALUE can appear on the left hand side of an assignment statement – they are also valid on the right hand side of an assignment statement.

Examples of Valid LValues (in bold)
Examples of Invalid LValues (in bold)

1000 NAME$ = "David"

1010 I% = 123 + 456

1020 DIM ARRAY(10)

1020 ARRAY(3+4) = 123.456
1000 "David" = NAME$

1010 123 + 456 = I%

An RVALUE is a value that is only valid on the right hand side of an assignment statement.. An RVALUE may be a variable, array element, constant or expression.

Variables are used by Mobile BASIC to manipulate and store values that vary over time. A variable is used within mobile basic by choosing a meaningful name and then assigning the values to the chosen name. As an example, your bank account will contain a certain amount of money that varies over time. A suitable name for this variable could simply be "BALANCE". Once you have decided on a suitable name you need to associate the variable with a value using an assignment statement:-

BALANCE = 1000.0

You will also need to access the current value of BALANCE at various times and this is done simply by using the variables name. For example, to print the value of BALANCE simply enter:-

PRINT BALANCE

At times you will add money to your account and remove money from your account. When this is done your balance needs to be updated accordingly. This is performed by creating simple expressions that take the current balance and adjusts it by adding or subtracting the appropriate value.

BALANCE = BALANCE + 100.0

BALANCE = BALANCE – 50

There are certain rules surrounding variable names. Variable names consist of alphabetic character followed by an optional series of alphanumeric characters. Variable names are case insensitive so "TEST", "test" and "TeSt" all refer to the same variable.

If a variable name is terminated by a "%" character then the variable will contain Integer Numbers only. Similarly, a variable name terminated by a "$" character is a String variable that can contains strings of text. e.g. "This is a string". Variable that are not terminated by "%" or "$" contain floating point values (e.g. 3.14).

Performance Tip: Try to use Integer variables (ending with %) wherever possible since they are directly supported by the Mobile Phone. Floating Point values need to be calculated within Mobile BASIC itself and take much longer to process.

Variable = Expression

Expression Examples:

Numeric Expression:
3 * 4

Numeric Expression:
A * ( B / 2 )

Relational Expression:
A < 3

Logical Expression:
(A < 3) AND (B > 4)

Printing Results

There is no point in having data in the computer if you cannot access its values. One of the most important methods of accessing the data is simply to print it onto the display. This operation is performed using the PRINT command.

The PRINT command takes a single integer, float or string expression and prints the value onto the screen. You will probably find that the string printing method gives you the most flexibility since it allows you to have several values on the same line as shown on line 1030.

1000 PRINT 25 * 2

1010 PRINT 100.0 / 4.0

1020 PRINT "STRING"

1030 PRINT "3 SQUARED = " + STR$(3*3)

Entering Data using INPUT

As well as getting data out of the computer it is equally important to get data into it as well. One of these methods is the INPUT command that takes a prompt and a variable to which the input is assigned.

1000 INPUT "Enter name: ",NAME$

1010 INPUT "Enter age: ",AGE%

1020 INPUT "Enter height: ",HEIGHT

The type of input is dependent on the variable specified to the INPUT statement. Line 1000 specifies a String variable, Line 1010 and Integer variable and Line 1020 a Float variable. If the user enters some data that cannot be converted to the type of the variable then an error will occur. The error should be trapped using the TRAP, which will be described later.

Control Flow

All Mobile BASIC programs start by executing the first line in the program followed by any subsequent lines. The following program calculates the number of years you have to normal retirement and prints the answer onto the display.

1000 INPUT "Age: ",AGE%

1010 YEARS%=65-AGE%

1020 PRINT "You have " + STR$(YEARS%) + " year(s) to retirement"

Every time you enter the "RUN" command you will be asked for an age and the program always responds with the number of remaining years.

1000 INPUT "Age: ",AGE%

1010 YEARS%=65-AGE%

1020 PRINT "You have " + STR$(YEARS%) + " year(s) to retirement"

1030 GOTO 1000

Once the program has printed the number of year to retirement the GOTO statement tells the program to resume at line 1000, whereupon the cycle starts again – this is an example of an infinite loop.

Now what happens if someone enters a value that is greater than the retirement age? If you entered 66 into the above program it would respond with "You have –1 years to retirement". Clearly the result doesn't make sense, we would now like to modify the program so that it prints the number of years to retirement or that this person has already retired. This process is called conditional execution and it is performed using the "IF" command.

The "IF" command consists of an expression followed by the "THEN" statement and then a list of statements that are to be executed if the expression was true. You can see an example of the "IF" command at line 1010 in the example below.

1000 INPUT "Age: ",AGE%

1010 IF (AGE% >= 65) THEN PRINT "Already retired": GOTO 1000

1020 YEARS%=65-AGE%

1030 PRINT "You have " + STR$(YEARS%) + " year(s) to retirement"

1040 GOTO 1000

You have probably noticed that there is no way of stopping these programs other than by selecting "STOP" from the menu. This is quite a drastic way of stopping a program and you should try to ensure that all your loops terminate by themselves so that the program stops in controlled manor. A simple way of adding this to the above program is to allow the user to enter a negative number when there are no more calculations required. You can see how this has been implemented by examining line 1020 and 1060 in the example below.

1000 INPUT "Age: ",AGE%

1010 IF (AGE% >= 65) THEN PRINT "Already retired": GOTO 1000

1020 IF (AGE% < 0) THEN GOTO 1060

1030 YEARS%=65-AGE%

1040 PRINT "You have " + STR$(YEARS%) + " year(s) to retirement"

1050 GOTO 1000

1060 PRINT "Program finished"

The following example illustrates a simple guess the number between 0 and 100 program. The RND() function returns an Integer value from -2147483648 to 2147483647. The program converts it into a value between 0 and 100 by first ensuring it is a positive number using ABS(). It then divides it by 100 and keeps the remainder (which will be between 0 and 100) using the MOD() function.

1000 N%=MOD(ABS(RND(0)),100)

1010 PRINT "Guess the number"

1020 INPUT "Guess: ",G%

1030 IF G%<N% THEN PRINT "Too low":GOTO 1020

1040 IF G%>N% THEN PRINT "Too high":GOTO 1020

1050 PRINT "Correct"

Looping

In the previous example you where introduced to loops using the GOTO command in conjunction with the IF command. Since loops are so frequently encountered within programs there is a special construct dedicated to just this tasks. This construct is called the FOR NEXT loop and it lets you execute a block of statements a given number of times. The FOR NEXT loop actually consists of two commands: The FOR loop itself which is used to initiate a loop and set its parameters and the NEXT command which is used to start the next iteration.

1000 FOR I%=1 TO 10

1010 PRINT I%

1020 NEXT I%

Line 1000 contains the FOR command which defines an integer control variable initial value (I%=1) – Please note that the control variable must be an INTEGER variable. Following the control variable definition you will see the "TO" command which is used to define the limit for the control variable (in this case 10). 

The control variable is just a normal variable and is accessible to the program just like any other. Line 1010 contains a "PRINT" statement that prints out the value of the control variable.

Line 1020 contains the "NEXT" statement which specifies the control variable of the "FOR" loop it is associated with. When the NEXT statement is executed it causes the control variable to be incremented by one. Once the control variable has been incremented it is compare with the limit value set in the "FOR" statement. If the control variable is still within limit then the loop is started again immediately following the "IF" statement. This continues until the control variable exceeds the limit.

The previous example illustrated how to count upwards from 1 to 10. Each time the next loop is executed then 1 is added to the control variable. This raises the question of how you would create a loop that counts down from 10 to 1. The answer is to use the "FOR" commands "STEP" attribute to specify a negative increment value.

1000 FOR I%=10 TO 1 STEP -1

1010 PRINT I%

1020 NEXT I%

If you do not specify the "STEP" attribute then it always defaults to 1. The previous example illustrated how you could specify a negative value in order to count down. You should also note that you can set this value to any value you like in order to count up in quantities other than 1.

1000 FOR I%=1 TO 100 STEP 10

1010 PRINT I%

1020 NEXT I%

This simply prints out 1, 11, 21, 31, 41, 51, 61, 71, 81 and finally 91. 101 isn't printed because it exceeds the limit set on the loop.

The following example calculates how much a sum of money will be worth after a number of years at a certain interest rate.

1000 INPUT "Principal: ",P

1010 INPUT "Rate %: ",R

1020 INPUT "Years: ",YEARS%

1030 FOR N%=1 TO YEARS%

1040 A=P*(1+R/100)^N%

1050 PRINT "YEAR: " + STR$(N%) + " AMOUNT: " + STR$(A)

1060 NEXT N%

Error Handling

There are events during the execution of your program that may cause errors to occur. It is a good idea to build in support for these errors so that corrective action may be taken. When we discussed the INPUT statement we simply said that an error could occur if the user entered data couldn't be converted to the type of the variable.  An error condition normally causes the program to immediately halt execution and print out an error message.

Look at the program below and consider what happens if the user entered the text "WJF*CQ" when prompted to enter their age. The INPUT statement was expecting an integer value but the user entered something that couldn't be sensibly converted to an integer … resulting in an error.

1000 INPUT "Enter name: ",NAME$

1010 INPUT "Enter age: ",AGE%

1020 INPUT "Enter height: ",HEIGHT

The program simply terminated with an error message and there was no option to continue other than by restarting the program again.

You should try to accommodate and handle errors that occur so that the program may continue execution. In many cases this simply involves telling the user that the value was incorrect and to re-enter the value. The TRAP statement is used to trap errors and it causes the program to resume at the specified line following an error.

1000 TRAP 2000

1010 INPUT "Enter name: ",NAME$

1020 INPUT "Enter age: ",AGE%

1030 INPUT "Enter height: ",HEIGHT

1040 PRINT "Thank you, " + NAME$

1050 END

2000 PRINT "Sorry you entered some bad data!"

2010 PRINT "Please re-enter"

2020 GOTO 1000

Arrays

Arrays are used to create tables, lists, sequences or collections of related values. An array can be thought of as a variable that contains more that one item of data. Each item of data is accessed or assigned by appending an integer index enclosed within parenthesis to the variable. In order to use a variable as an array you must dimension the variable with the DIM command. As an example lets suppose you have a Car that can be purchased in Five colours: Red, Green, Blue, Black and White. The first thing that must be done is to declare COLOUR$ as an array of five strings.

1000 DIM COLOUR$(5)

Now that we have our array we can assign our three colours to it as shown below: -

1000 DIM COLOUR$(5)

1010 COLOUR$(0) = "Red"

1020 COLOUR$(1) = "Green"

1030 COLOUR$(2) = "Blue"

1040 COLOUR$(3) = "Black"

1050 COLOUR$(4) = "White"

Note that the indexes for the 5 items are based from 0 and not 1.

The contents of the array can be printed using the FOR loop specified in the previous section.

1000 DIM COLOUR$(5)

1010 COLOUR$(0) = "Red"

1020 COLOUR$(1) = "Green"

1030 COLOUR$(2) = "Blue"

1040 COLOUR$(3) = "Black"

1050 COLOUR$(4) = "White"

1060 FOR I%=0 TO 4

1070 PRINT COLOUR$(I%)

1080 NEXT I%

Read and Data

In the previous example you will have noticed that each array value needs to be set individually. In those example we assigned them explicitly using an assignment statement. A more compact way of performing initialisation is to READ the values from DATA statements. In the following example, Line 1020 introduces the READ command that reads a value and assigns it to the specified variables array index. The READ command gets it data from the current DATA statement which by default is the first DATA statement found in the program.

1000 DIM COLOUR$(5)

1010 FOR I%=0 TO 4

1020 READ COLOUR$(I%)

1030 NEXT I%

1040 FOR I%=0 TO 4

1050 PRINT COLOUR$(I%)

1060 NEXT I%

2000 DATA Red,Green,Blue,Black,White

We can actually make the above program more flexible by allowing you to alter the number of colours simply by updating the DATA statement. This is done simply by inserting the number as the first item in the DATA statement.

1000 READ NC%

1010 DIM COLOUR$(NC%)

1020 FOR I%=0 TO NC%-1

1030 READ COLOUR$(I%)

1040 NEXT I%

1050 FOR I%=0 TO NC%-1

1060 PRINT COLOUR$(I%)

1070 NEXT I%

2000 DATA 5,Red,Green,Blue,Black,White

A Yellow colour can simply be added by updating the DATA statement to read: -

2000 DATA 6,Red,Green,Blue,Black,White,Yellow

Each DATA statement consists of a comma separated list of values that are read sequentially. When the last item has been read from the current DATA statement then the system moves to the next DATA statement in the program. There are times when you don't want to read DATA in sequential order and this is where the RESTORE command can be used.

1000 RESTORE 2010

1010 READ S$

1020 PRINT S$

2000 DATA One

2010 DATA Two

2020 DATA Three

The RESTORE statement on line 1000 causes the DATA items to be read starting from Line 2010. You will find that the above program simply prints "Two" – the first data value has been ignored.

If your program executes a READ statement when there is no more data to be read then Mobile BASIC issues an OUT OF DATA error which should be trapped using the TRAP command discussed previously.

String Handling

1000 A$="Hello World"

1010 PRINT A$

1020 PRINT LEFT$(A$,5)

1030 PRINT RIGHT$(A$,5)

1040 PRINT MID$(A$,7,3) + RIGHT$(A$,1)

1050 PRINT "LEN="+STR$(LEN(A$))

1060 PRINT RIGHT$(A$,5)+" "+LEFT$(A$,5)

The following code is a string sort

1000 DIM A$(5)

1010 A$(0)="PETER"

1020 A$(1)="DAVID"

1030 A$(2)="MARY"

1040 A$(3)="JOHN"

1050 A$(4)="CATHERINE"

1060 SWITCHED%=0

1070 FOR I%=0 TO 3

1080 IF A$(I%)<A$(I%+1) THEN GOTO 1130

1090 T$=A$(I%)

1100 A$(I%)=A$(I%+1)

1110 A$(I%+1)=T$

1120 SWITCHED%=1

1130 NEXT I%

1140 IF SWITCHED%=1 THEN GOTO 1060

1150 FOR I%=0 TO 4

1160 PRINT A$(I%)

1170 NEXT I%

1180 END

Input / Output

There are two forms of I/O supported by Mobile BASIC

Local Record Based Files

When a file is first opened it default to reading or writing record 1.

A new record is only selected by using the POINT command.

You should try to keep the records reasonably small since the whole record is loaded into memory when it is access. Very large records will consequently have an adverse effect on the amount of memory available for other tasks.

Remote Internet Services

Connect through the http protocol

Talk to remote Java Servlets or other CGI type programs

I/O to remote services is slightly different than the other files and is in two phases – The output phase and the input phase. Phase 1 (WRITE ONLY) allows you to write data to the channel. Phase 2 (READ ONLY) is started as soon as you attempt to read data from the channel. You cannot return to Phase 1 without re-opening the channel.

NOTE and POINT are not supported for Internet Based Files and Services.

1000 OPEN #1,"test2.dat","OUTPUT"

1010 FOR I%=1 TO 10");

1020 POINT #1,I%

1030 PUT #1,I%

1040 PRINT #1,123456789

1050 PRINT #1,1.234*I%

1060 PRINT #1,"Hello World"

1070 NEXT I%

1080 CLOSE #1

2000 OPEN #1,"test2.dat","INPUT"

2010 REC%=1

2020 POINT #1,REC%

2030 TRAP 3000

2040 GET #1,I%

2050 INPUT #1,J%

2060 INPUT #1,F

2070 INPUT #1,S$

2080 PRINT STR$(I%)+" "+STR$(F)+" "+S$

2090 REC%=REC%+1

2100 GOTO 2020

3000 PRINT "EOF"

3010 CLOSE #1

Graphics

1000 CLS

1010 PLOT 5,5

1020 DRAWLINE 5,5,25,25*3

1030 SETCOLOR 255,0,0

1040 FILLRECT 40,30,30,15

1050 DRAWRECT 35,25,40,25

1060 FILLROUNDRECT 30,20,50,35,10,10

1070 DRAWROUNDRECT 25,15,60,45,10,10

1080 SETCOLOR 0,0,255

1090 FILLARC 30,30,50,50,60,-60

1100 SETCOLOR 0,255,0

1110 H$="Press Fire"

1120 XPOS%=SCREENWIDTH(0)-STRINGWIDTH(H$)

1130 YPOS%=SCREENHEIGHT(0)-STRINGHEIGHT(H$)

1140 DRAWSTRING H$,XPOS%,YPOS%

1150 IF FIRE(0)=0 THEN GOTO 1150

1160 BLIT 0,0,SCREENWIDTH(0),SCREENHEIGHT(0),20,0

1170 END

User Interfaces

1000 DIM A$(3)

1010 A$(0)="Alloy Wheels"

1020 A$(1)="Leather"

1030 A$(2)="CD Player"

1040 I%=CHOICEFORM("Choice Form", "Continue", "Cancel", "My Choice", A$, 0)

1050 PRINT I%

1060 I%=CHOICEFORM("Options", "Continue", "Cancel", "Optional Extras", A$, 1)

1070 PRINT I%

2000 DIM J%(2)

2010 J%(0)=DAYS(0)

2020 J%(1)=MILLISECONDS(0)

2030 I%=DATEFORM("Date Form", "Continue", "Cancel", "Today", J%, 0)

2035 PRINT I%

2040 I%=DATEFORM("Date Form2", "Continue", "Cancel", "My Date2", J%, 0)

2050 PRINT I%

3000 T$="Default Text"

3010 I%=EDITFORM("Edit Form", "Continue", "Cancel", "Description", T$, 40, 0)

3020 PRINT STR$(I%) + ": " + T$

3030 PRINT I%

3040 I%=EDITFORM("Edit Form", "Continue", "Cancel", "Password", T$, 40, 1)

3050 PRINT T$

3060 PRINT I%

3070 T$="1234"

3080 I%=EDITFORM("Edit Form", "Continue", "Cancel", "Numeric", T$, 40, 2)

3090 PRINT T$

3100 PRINT I%

4000 I% = 5

4010 I%=GAUGEFORM("Gauge Form", "Continue", "Cancel", "Volume", 10, I%, 1)

4020 PRINT I%

4030 IF I%<>-1 THEN GOTO 4010

5000 I%=MESSAGEFORM("Message Form", "Continue", "Cancel", "Info: ", "Mobile BASIC contains several build-in forms including:- CHOICEFORM(), EDITFORM(), DATEFORM(), GAUGEFORM() and MESSAGEFOR()")

5010 PRINT I%

Mathematical Calculations

1000 PRINT "SQR(81.0)=" + STR$(SQR(81.0))

1010 PRINT "3.0 ^ 2.0=" + STR$(3.0 ^ 2.0)

1020 PRINT "LOG(97.0)=" + STR$(LOG(97.0))

1030 PRINT "EXP(4.317488)" + STR$(EXP(4.317488))

1040 DEG

1050 SIN45=SIN(45.0)

1060 COS45=COS(45.0)

1070 TAN45=TAN(45.0)

1080 PRINT "ASIN(" + STR$(SIN45) + ")=" + STR$(ASIN(SIN45))

1090 PRINT "ACOS(" + STR$(COS45) + ")=" + STR$(ACOS(COS45))

1100 PRINT "ATAN(" + STR$(TAN45) + ")=" + STR$(ATAN(TAN45))

1110 RAD

1120 ANG30=30/57.29578

1130 PRINT ANG30

1140 SIN30=SIN(ANG30)

1150 COS30=COS(ANG30)

1160 TAN30=TAN(ANG30)

1170 PRINT "ASIN(" + STR$(SIN30) + ")=" + STR$(ASIN(SIN30))

1180 PRINT "ACOS(" + STR$(COS30) + ")=" + STR$(ACOS(COS30))

1190 PRINT "ATAN(" + STR$(TAN30) + ")=" + STR$(ATAN(TAN30))

Sprite Facilities

This program draws a red and a green cross onto the screen. These crosses are then loaded into Graphics Elements named "Cursor0" and "Cursor1". The screen is cleared and sprite variables are initialised.

The main loop starts at line 1130 which associates the named sprite "Cursor" with one of the named GELs. Line 1140 Moves the sprite to the appropriate location. Line 1150 and 1160 perform throttling which ensures that the loop doesn't execute to quickly. Line 1170-1230 handle user input. LEFT, RIGHT, UP and DOWN cause the cursor to move in that direction. The FIRE button causes the sprite to switch between the two graphics elements.

1000 CLS

1010 SETCOLOR 255,0,0

1020 DRAWLINE 0,0,15,15

1030 DRAWLINE 15,0,0,15

1040 SETCOLOR 0,255,0

1050 DRAWLINE 20,0,35,15

1060 DRAWLINE 35,0,20,15

1070 GELLOAD "Cursor0",0,0,15,15

1080 GELLOAD "Cursor1",20,0,15,15

1090 CLS

1100 SPRITEX%=0

1110 SPRITEY%=0

1120 GELNUM%=0

1130 SPRITEGEL "Cursor","Cursor" + STR$(GELNUM%)

1140 SPRITEMOVE "Cursor",SPRITEX%,SPRITEY%

1150 MS%=MILLISECONDS(0)+50

1160 IF MILLISECONDS(0)<MS% THEN GOTO 1160

1170 IF LEFT(0)<>0 THEN SPRITEX%=SPRITEX%-1

1180 IF RIGHT(0)<>0 THEN SPRITEX%=SPRITEX%+1

1190 IF UP(0)<>0 THEN SPRITEY%=SPRITEY%-1

1200 IF DOWN(0)<>0 THEN SPRITEY%=SPRITEY%+1

1210 IF FIRE(0)=0 THEN GOTO 1140

1220 GELNUM%=1-GELNUM%

1230 GOTO 1130

Game Features
Time and Date Example

1000 DAYS%=DAYS(0)

1010 MS%=MILLISECONDS(0)

1020 Y%=YEAR(DAYS%, MS%)

1030 M%=MONTH(DAYS%, MS%)

1040 D%=DAY(DAYS%, MS%)

1050 PRINT STR$(D%)+"/"+STR$(M%)+"/"+STR$(Y%)

1060 H%=HOUR(DAYS%, MS%)

1070 M%=MINUTE(DAYS%, MS%)

1080 S%=SECOND(DAYS%, MS%)

1090 MS%=MILLISECOND(DAYS%, MS%)

1100 PRINT STR$(H%)+":"+STR$(M%)+":"+STR$(S%)+"."+STR$(MS%)

Communication with an Internet Server

To simplify data transfer we suggest that you send and received your data in string format. i.e. Where floats and integers are to be sent to a server they should first be converted to a string using the STR$() function. Likewise numeric values in strings returned from the server should be converted into Mobile BASIC format using the VAL() function.

Java Servlet – (HelloServer Servlet)

The following is an example of a Network Resource. The server simply reads a name and response with 'Hello: "name"'.  The example is implemented as a Java Servlet suitable for using Web servers supporting Java Servlets.

The Mobile BASIC program to talk to the HelloServer Java Servlet
1000 OPEN #0,"http://www.mobilebasic.com/servlet/HelloServer","OUTPUT"

1010 INPUT "Name: ",N$

1020 PRINT #0,N$

1030 INPUT #0,A$

1040 PRINT A$

1050 CLOSE #0

import java.io.*;

import javax.servlet.*;

import javax.servlet.http.*;

public class HelloServer extends HttpServlet

{

  public void doGet(HttpServletRequest request, HttpServletResponse response)

    throws ServletException ,IOException

  {

  }

  public void doPost(HttpServletRequest request, HttpServletResponse response)

    throws ServletException ,IOException

  {

    InputStream inputStream = request.getInputStream();

    DataInputStream dataInputStream = new DataInputStream(inputStream);

    String name = dataInputStream.readUTF();

    OutputStream outputStream = response.getOutputStream();

    DataOutputStream dataOutputStream = new DataOutputStream(outputStream);

    dataOutputStream.writeUTF("Hello \"" + name + "\"");

  }

}

Sending Data to Server

Data Type
Mobile BASIC Command to Send the Data


Java Servlet Command to Read the Data

Byte
PUT #n, byte
byte = dataInputStream.readByte()

Integer
PRINT #n, int%
int I = dataInputStream.readInt()

Float – Mobile BASIC uses its own floating point format because floats are unsupported by Java MIDlets.
PRINT #n, float


int I = dataInputStream.readInt()

String
PRINT #n, "String"
String s = dataInputStream.readUTF()

Reading Data from Server

Data Type
Java Servlet Command to Send the Data
Mobile BASIC Command to Read the Data

Byte
dataOutputStream.writeByte(intValue)
GET #1, intVariable

Integer
dataOutputStream.writeInt(intValue)
INPUT #1, intVariable

Float – Mobile BASIC uses its own floating point format because floats are unsupported by Java MIDlets.
dataOutputStream.writeInt(intValue)
INPUT #1, floatVariable

String
dataOutputStream.writeUTF("String")
INPUT #1, stringVariable

PHP Script – HelloServer.php

Things to note Mobile BASIC sends strings in UTF8 format. If you are simply using ASCII data then it is simply a two byte byte count (hi byte, lo byte) immediately before the ASCII string. The string length can be determined by "len = hi * 256 + lo"

The Mobile BASIC program to talk to the HelloServer PHP Script

1000 OPEN #0,"http://www.mobilebasic.com/HelloServer.php","OUTPUT"

1010 INPUT "Name: ",N$

1020 PRINT #0,N$

1030 INPUT #0,A$

1040 PRINT A$

1050 CLOSE #0

<script language="PHP">

$rawdata = $HTTP_RAW_POST_DATA;

$hi = ord($rawdata[0]);

$lo = ord($rawdata[1]);

$len = $hi * 256 + $lo;

$response = "Hello \"".substr($rawdata,2,$len)."\"";

$len=strlen($response);

$hi = $len / 256;

$lo = $len % 256;

echo chr($hi).chr($lo).$response

</script>

Example Programs

Dice Program

1000 W%=SCREENWIDTH(0)

1010 H%=SCREENHEIGHT(0)

1020 N%=MOD(ABS(RND(0)),6)+1

1030 RESTORE 9000+N%

1040 CLS

1050 READ NDOTS%

1060 FOR I%=1 TO NDOTS%

1070 READ XPOS

1080 READ YPOS

1090 XPOS=XPOS*W%

1100 YPOS=YPOS*H%

1110 FILLROUNDRECT XPOS-3,YPOS-3,7,7,3,3

1120 NEXT I%

1130 IF FIRE(0)=0 THEN GOTO 1130

1140 IF FIRE(0)=1 THEN GOTO 1140

1150 GOTO 1020

9001 DATA 1,0.5,0.5

9002 DATA 2,0.25,0.5,0.75,0.5

9003 DATA 3,0.25,0.25,0.5,0.5,0.75,0.75

9004 DATA 4,0.25,0.25,0.75,0.25,0.25,0.75,0.75,0.75

9005 DATA 5,0.25,0.25,0.75,0.25,0.25,0.75,0.75,0.75,0.5,0.5

9006 DATA 6,0.25,0.25,0.5,0.25,0.75,0.25,0.25,0.75,0.5,0.75,0.75,0.75

